

任务 2.2 用变量存储学生信息




图 2.3 用变量存储学生信息

在执行应用程序期间，可以使用变量来临时存储数据。变量有名称和数据类型，名称是用于引用变量所包含的值的词，而数据类型则用于确定变量能够存储的数据的种类，可以将变量看作计算机内存中存放未知数据的所在。在本任务中，将声明一些变量，分别用于存储学生的姓名、性别、身高、出生日期、入学日期、是否共青团员和手机号码等信息，并要求在窗体上显示这些信息。程序运行结果如图 2.3 所示。

任务目标

- 理解 Visual Basic 中变量的概念和作用。
- 掌握使用 Dim 语句声明变量的方法。
- 掌握使用赋值语句对变量赋值的方法。
- 掌握两种注释语句的使用方法。

任务分析

使用变量存储学生信息主要有 3 个步骤：使用 Dim 语句声明变量的名称和数据类型；使用赋值语句指定在变量中存储的值；使用 Print 方法在窗体上显示变量的值。

任务实施

- (1) 在 Visual Basic 6.0 集成开发环境中创建一个标准 EXE 工程。
- (2) 在属性窗口中将窗体 Form1 的 Caption 属性值修改为“用变量存储学生信息”。
- (3) 在窗体 Form1 的代码窗口中编写以下事件过程：

```
Private Sub Form_Click()  
    Rem 声明一些变量，用于存储学生信息  
    Dim Name As String      '姓名  
    Dim Gender As String    '性别  
    Dim Height As Single    '身高  
    Dim Birthdate As Date   '出生日期  
    Dim EntryDate As Date   '入学日期  
    Dim IsCY As Boolean      '是否共青团员
```

```

Dim Mobile As String      '手机号码

Rem 对变量赋值
Name = "张志明"
Gender = "男"
Height = 1.78
Birthdate = #5/18/2002#
EntryDate = #8/8/2017#
IsCY = True
Mobile = "18803711618"
Const hr = "-----"
Rem 在窗体上显示学生信息
Cls
Print
Print Spc(26); "学生信息"
Print Tab(6); hr; hr
Print Tab(12); "姓名: "; Spc(10); Name
Print Tab(12); "性别: "; Spc(10); Gender
Print Tab(12); "身高: "; Spc(9); Height; "m"
Print Tab(12); "出生日期: "; Spc(6); Birthdate
Print Tab(12); "入学日期: "; Spc(6); EntryDate
Print Tab(12); "是否共青团员: "; Spc(2); IsCY
Print Tab(12); "手机号码: "; Spc(6); Mobile
Print Tab(6); hr; hr

End Sub

```

(4) 将窗体文件保存为 Form02-02.frm，工程文件保存为工程 02-02.vbp。

程序测试

- (1) 按下 F5 键，以运行程序。
- (2) 在程序运行期间，使用鼠标单击程序窗口以查看输出结果。
- (3) 单击窗口右上角的关闭按钮，以结束程序运行。

相关知识

1. 变量

变量就是命名的存储单元位置，包含在程序执行阶段修改的数据。每个变量都有名称，并且在范围内可被唯一识别；变量的名称必须以字母开头，且中间不能包含英文句点(.)或类型说明符(如\$)，最大长度不能超过 255 个字符。在声明变量时可以指定其数据类型，也可以不指定。借助变量名就可以访问内存中的数据。变量在使用前一般需要预先声明，声明变量就是将变量的有关信息事先告诉编译系统。变量名用于识别变量在内存中的位置，变量的数据类型




图 2.4 设置编辑器选项

型指定其占用内存空间的大小。

如果希望在程序中强制显式声明所有变量，则可以在模块的声明段中加入 Option Explicit 语句，也可以在“工具”下拉菜单中选择“选项”命令，在弹出的“选项”对话框中单击“编辑器”选项卡，勾选“要求变量声明”复选框，如图 2.4 所示。这样将会在任何新建模块中自动插入 Option Explicit 语句。

在 Visual Basic 中，可以使用 Dim 语句来声明变量的数据类型并分配存储空间，语法格式如下：

```
Dim|Private|Static|Public <变量名> [As 数据类型] [, <变量名> [As 数据类型]]
```

其中，“变量名”的命名必须遵循标识符的命名规则；“数据类型”用于声明变量的数据类型或对象类型。

Visual Basic 中变量的类型、作用域、声明位置及使用关键字如表 2.2 所示。

表 2.2 Visual Basic 中变量的类型、作用域、声明位置及使用关键字

变 量 类 型	作 用 域	声 明 位 置	使用关键字
局部变量	过程	过程中	Dim 或 Static
模块变量	窗体模块或标准模块	模块的声明部分	Dim 或 Private
全局变量	整个应用程序	标准模块的声明部分	Public

2. 变量的作用域

变量的作用域指的是变量的有效范围。为了正确地使用变量的值，应当明确在程序的什么地方可以访问该变量。在 Visual Basic 中，变量按照作用域的不同可以分为局部变量、模块变量和全局变量，其中模块变量分为窗体模块变量和标准模块变量。

局部变量只有在声明它们的过程中才能被识别，它们又被称为过程级变量。

模块变量对该模块的所有过程都可用，但是对其他模块的代码不可用。可以在模块顶部的声明段中使用 Private 或 Dim 关键字来声明模块变量。

全局变量是使用 Public 关键字声明的变量，其值可以用于应用程序的所有过程。与所有模块变量相同，也可以在模块顶部的声明段中来声明全局变量。

在过程执行结束后，局部变量的值不能被保留下来。在每次重新执行过程时，局部变量会被重新初始化。如果希望在该过程结束之后还能继续保持过程中局部变量的值，则应该使用 Static 关键字将这个变量声明为静态变量。这样，即使过程结束，该静态变量的值也仍然被保留着。

在使用声明语句声明一个变量后，Visual Basic 会自动将数值型的变量赋初值为 0，将字符串型或 Variant 数据类型的变量赋初值为空字符串，将布尔型的变量赋初值为 False，将日期型的变量赋初值为 00:00:00（如果系统为 12 小时制，则会有所不同）。

在 Visual Basic 中，也允许变量不经过声明就直接使用，这种情况被称为隐式声明。所有隐式声明的变量都是 Variant 数据类型的，并默认为局部变量。

3. 赋值语句

如果想要将表达式的值赋给变量或对象的属性，则可以使用赋值语句来实现，语法格式如下：

```
[Let] 变量名 | 属性名 = 表达式
```

其中，Let 关键字通常被省略；变量或属性的名称应遵循标准变量的命名规则；“表达式”用于指定要赋给变量或属性的值。

只有当表达式的数据类型与变量的数据类型兼容时，该表达式的值才可以被赋给变量或属性。不能将字符串表达式的值赋给数值变量，也不能将数值表达式的值赋给字符串变量。如果这样做，则会在编译时出现错误。

变量的数据类型和表达式的数据类型必须一致。如果两者同为数值型但是精度不一样，则系统会强制将表达式的值转换为变量所要求的精度。

赋值号(=)表示将表达式的值赋给变量，与数学中的等号的意义不同。Let 关键字表示赋值，通常省略。例如，在 x=8=9 语句中，左侧的“=”为赋值号，右侧的“=”为比较运算符，这个语句的作用是将关系表达式 8=9 的结果赋给变量 x，因此 x 的值为 0 (False)。

如果想要将对象引用赋给变量或属性，则不能使用赋值语句，而必须使用 Set 语句。

赋值语句兼有计算与赋值双重功能，它首先计算赋值号右侧“源操作符”的值，然后把结果赋给赋值号左侧的“目标操作符”。

如果想要把多个赋值语句放在同一行，则各个语句之间必须使用冒号隔开。示例如下：

```
a=3:b=4:c=5。
```

4. 注释语句

注释语句是为了方便阅读而对程序进行的说明，对程序运行没有影响。

Visual Basic 提供了两种注释语句，语法格式如下：

```
Rem 注释文本  
'注释文本
```

注释语句可以单独占一行，也可以放在其他语句的行尾。当使用 Rem 关键字时，该关键字与注释文本之间需要添加一个空格；当放在其他语句的行尾时，需要使用冒号(:) 隔开。当使用撇号(') 时，则不必加冒号。

5. 结束语句

在 Visual Basic 中有两个结束语句，语法格式如下：

```
End  
Unload <对象名称>
```

其中，End 语句用于结束正在运行的程序，它提供了一种强行终止程序的方法；Unload 语句用于从内存中卸载窗体或控件，在卸载窗体前，依次发生窗体的 QueryUnload 和 Unload 事件过程。

6. 语句格式

当在 Visual Basic 中编写程序语句时，需要注意以下规则。

- 不区分大小写。Visual Basic 对关键字有自动转换大小写功能。例如，在输入 Print 语句时，不论输入 PRINT 还是 print，Visual Basic 都会将其转换为 Print。
- 标点符号都需要在英文状态下输入。
- 语句以 Enter 键结束。在一般情况下，要求“一句一行”。不过也可以使用复合语句，即将多个语句写在同一行上，但是每个语句之间必须使用冒号(:)来连接。另外，当一行代码很长时，可以使用“空格+下画线”来续行。