

任务 4.3 创建简单选课系统

在本任务中，将创建一个简单的选课系统。当从顶部的组合框中选择不同专业时，左侧列表框将自动列出相关的课程，可以选择所需要的课程并添加到右侧列表框中；也可以从右侧列表框中移除已经选定的课程，如图 4.9 所示。当完成课程选择并单击“确定”按钮时，将弹出另一个窗体，在此窗体中列出选课结果，如图 4.10 所示。




图 4.9 选择专业和课程




图 4.10 查看选课结果

任务目标



- 掌握列表框控件的应用。
- 掌握组合框控件的应用。

任务分析

列表框和组合框都可以用于显示项目列表。在本任务中，通过组合框来显示专业名称列表，当从该组合框中选择某个专业时，会在左侧列表框中显示与该专业相关的可选课程。想要实现这一点，可以对“专业”组合框的 Click 事件进行编程，通过 ListIndex 属性来获取当前所选择的专业，并根据专业不同调用 AddItem 方法动态加载项目。通过模块级数组来存储这些项目，并在选课窗体的 Load 事件过程中将字符串拆分成一个数组，然后把每个数组元素添加到列表框中。

任务实施

- (1) 在 Visual Basic 6.0 集成开发环境中创建一个标准 EXE 工程。
- (2) 将窗体 Form1 命名为 frmSelect，然后将其 Caption 属性值设置为“选课系统”。
- (3) 在窗体 frmSelect 上添加以下控件。

- 添加标签控件并命名为 lblMajor, 然后将其 Caption 属性值设置为 “专业(&M):”。
- 在工具箱窗口中单击 ComboBox 控件图标, 在标签控件 lblMajor 的右侧绘制一个组合框控件, 然后将其命名为 cmbMajor。
- 添加标签控件并命名为 lblCourse, 将其 Caption 属性值设置为 “可选课程(&A):”。
- 在工具箱窗口中单击 ListBox 控件图标, 在标签控件 lblCourse 的下方绘制一个列表框控件, 然后将其命名为 lstCourse。
- 在列表框控件 lstCourse 的右侧添加命令按钮控件并命名为 cmdAdd, 然后将其 Caption 属性值设置为 >, ToolTipText 属性值设置为 “添加选中的课程”。
- 在命令按钮控件 cmdAdd 下方添加命令按钮控件并命名为 cmdAddAll, 然后将其 Caption 属性值设置为 >>, ToolTipText 属性值设置为 “添加全部课程”。
- 在命令按钮控件 cmdAddAll 的下方添加命令按钮控件并命名为 cmdRemove, 然后将其 Caption 属性值设置为 <, ToolTipText 属性值设置为 “移除选中的课程”。
- 在命令按钮控件 cmdRemove 的下方添加命令按钮控件并命名为 cmdRemoveAll, 然后将其 Caption 属性值设置为 <<, ToolTipText 属性值设置为 “移除全部课程”。
- 添加标签控件并命名为 lblSelectedCourse, 然后将其 Caption 属性值设置为 “已选课程(&S):”。
- 在标签控件 lblSelectedCourse 的下方添加列表框控件并命名为 lstSelectedCourse。
- 在窗体底部添加命令按钮控件并命名为 cmdOK, 然后将其 Caption 属性值设置为 “确定(&O):”, Default 属性值设置为 True。
- 在命令按钮控件 cmdOK 的右侧添加命令按钮控件并命名为 cmdCancel, 然后将其 Caption 属性值设置为 “取消”, Cancel 属性值设置为 True。

(4) 在窗体 frmSelect 的代码窗口中声明一些窗体级变量, 程序代码如下:

```
'声明窗体级变量
Private sMajor As String, sCourse1 As String, sCourse2 As String, sCourse3 As String
Private aMajor() As String, aCourse1() As String, aCourse2() As String, Course3() As String
```

(5) 在窗体 frmSelect 的代码窗口中编写一个名称为 ListFill 的通用过程, 程序代码如下:

```
.....
'过程名: ListFill
'参数: oName 用于指定一个列表框或组合框, aItem 用于指定一个数组名
'功能: 使用数组 aItem 中的元素来填充由 oName 指定的列表框或组合框
.....
Sub ListFill(ByVal oName As Control, aItem() As String)

    Dim i As Integer
```



```

For i = 0 To UBound(aItem)
    oName.AddItem aItem(i)
Next

End Sub

```

(6) 在窗体 `frmSelect` 的代码窗口中编写该窗体的 `Load` 事件过程，程序代码如下：

```

'当加载窗体 Form1 时执行以下事件过程
Private Sub Form_Load()

    sMajor = "计算机应用技术,多媒体应用技术,计算机网络技术"
    sCourse1 = "计算机应用基础,计算机网络基础,计算机组装与维护,Visual Basic 程序设计,Dreamweaver 网页制作,Access 数据库应用基础,SQL Server 数据库,信息管理系统开发实训"
    sCourse2 = "计算机应用基础,美术设计,多媒体技术应用,Photoshop 图像处理,Flash 二维动画制作,Maya 三维动画制作,Premiere 影视编辑,After Effects 影视特效与合成"
    sCourse3 = "计算机应用基础,计算机网络基础,操作系统与网络服务器管理,网络综合布线技术,路由器交换机配置与管理,小型局域网搭建,ASP 动态网站开发,中小型网站建设与管理"

    '利用 Split 函数将字符串拆分成数组
    aMajor = Split(sMajor, ",")
    ListFill cmbMajor, aMajor
    cmbMajor.Text = cmbMajor.List(0)
    aCourse1 = Split(sCourse1, ",")
    aCourse2 = Split(sCourse2, ",")
    aCourse3 = Split(sCourse3, ",")

    '使用数组 aCourse1 中的元素来填充“可选课程”列表框
    ListFill lstCourse, aCourse1

End Sub

```

(7) 在窗体 `frmSelect` 的代码窗口中编写组合框控件 `cmbMajor` 的 `Click` 事件过程，程序代码如下：

```

'当从“专业”组合框中选择一个专业时执行以下事件过程
Private Sub cmbMajor_Click()

    '清除“可选课程”列表框
    lstCourse.Clear
    '清除“已选课程”列表框
    lstSelectedCourse.Clear
    '根据选择的专业的不同，使用不同的课程来填充“可选课程”列表框
    Select Case cmbMajor.ListIndex
        Case 0
            ListFill lstCourse, aCourse1
        Case 1

```

```

        ListFill lstCourse, aCourse2
    Case 2
        ListFill lstCourse, aCourse3
    End Select

```

```
End Sub
```

(8) 在窗体 `frmSelect` 的代码窗口中编写命令按钮控件 `cmdAdd` 的 Click 事件过程，程序代码如下：

'当单击 ">" 按钮时执行以下事件过程

```
Private Sub cmdAdd_Click()
```

```

    If lstCourse.ListCount > 0 And lstCourse.ListIndex <> -1 Then
        lstSelectedCourse.AddItem lstCourse.List(lstCourse.ListIndex)
        lstCourse.RemoveItem lstCourse.ListIndex
    ElseIf lstCourse.ListCount > 0 And lstCourse.ListIndex = -1 Then
        MsgBox "请选择要添加的课程!", vbInformation + vbOKOnly, Me.Caption
    End If

```

```
End Sub
```

(9) 在窗体 `frmSelect` 的代码窗口中编写命令按钮控件 `cmdAddAll` 的 Click 事件过程，程序代码如下：

'当单击 ">>" 按钮时执行以下事件过程

```
Private Sub cmdAddAll_Click()
```

```
    Dim i As Integer
```

```

    If lstCourse.ListCount > 0 Then
        For i = lstCourse.ListCount - 1 To 0 Step -1
            lstSelectedCourse.AddItem lstCourse.List(i)
            lstCourse.RemoveItem 0
        Next
    End If

```

```
End Sub
```

(10) 在窗体 `frmSelect` 的代码窗口中编写命令按钮控件 `cmdRemove` 的 Click 事件过程，程序代码如下：

'当单击 "<" 时执行以下事件过程

```
Private Sub cmdRemove_Click()
```

```

    If lstSelectedCourse.ListCount > 0 And lstSelectedCourse.ListIndex <> -1
Then
        lstCourse.AddItem lstSelectedCourse.List(lstSelectedCourse.ListIndex)
        lstSelectedCourse.RemoveItem lstSelectedCourse.ListIndex
    End If

```



```
End If
```

```
End Sub
```

(11) 在窗体 `frmSelect` 的代码窗口中编写命令按钮控件 `cmdRemove` 的 Click 事件过程，程序代码如下：

```
'当单击 "<<" 按钮时执行以下事件过程
```

```
Private Sub cmdClear_Click()
```

```
Dim i As Integer
```

```
If lstSelectedCourse.ListCount > 0 Then
```

```
For i = lstSelectedCourse.ListCount - 1 To 0 Step -1
```

```
lstCourse.AddItem lstSelectedCourse.List(0)
```

```
lstSelectedCourse.RemoveItem 0
```

```
Next
```

```
End If
```

```
End Sub
```

(12) 在窗体 `frmSelect` 的代码窗口中编写命令按钮控件 `cmdCancel` 的 Click 事件过程，程序代码如下：

```
'当单击 "取消" 按钮时执行以下事件过程
```

```
Private Sub cmdCancel_Click()
```

```
'卸载选课结果窗体和选课窗体
```

```
Unload frmResults
```

```
Unload Me
```

```
End Sub
```

(13) 在窗体 `frmSelect` 的代码窗口中编写命令按钮控件 `cmdOK` 的 Click 事件过程，程序代码如下：

```
'当单击 "确定" 按钮时执行以下事件过程
```

```
Private Sub cmdOK_Click()
```

```
If lstSelectedCourse.ListCount > 0 Then
```

```
Me.Hide
```

```
frmResults.Show
```

```
Else
```

```
MsgBox "你还没有选择任何课程!", vbInformation + vbOKOnly, "提示信息"
```

```
End If
```

```
End Sub
```

(14) 在窗体 `frmSelect` 的代码窗口中编写列表框控件 `lstCourse` 的 DblClick 事件过程，程序代码如下：

'当在“可选课程”列表框中双击一门课程时执行以下事件过程

```
Private Sub lstCourse_DblClick()
```

'调用命令按钮控件 cmdAdd 的 Click 事件过程

```
cmdAdd_Click
```

```
End Sub
```

(15) 在窗体 frmSelect 的代码窗口中编写列表框控件 lstSelectedCourse 的 DblClick 事件过程, 程序代码如下:

'当在“已选课程”列表框中双击一门课程时执行以下事件过程

```
Private Sub lstSelectedCourse_DblClick()
```

'调用命令按钮控件 cmdRemove 的 Click 事件过程

```
cmdRemove_Click
```

```
End Sub
```

(16) 在当前工程中添加一个新窗体并命名为 frmResults, 然后将该窗体的 Caption 属性值设置为“选课结果”。

(17) 在窗体 frmResults 上添加以下控件。

- 添加标签控件并命名为 lblResult, 然后将其 Caption 属性值清空。
- 在标签控件 lblResult 的下方添加文本框控件并命名为 txtResult, 然后将其 MultiLine 属性值设置为 True, Enabled 属性值设置为 False, ScrollBars 属性值设置为 3。
- 添加命令按钮控件并命名为 cmdPrevious, 然后将其 Caption 属性值设置为“上一步(&P)”。
- 添加命令按钮控件并命名为 cmdConfirm, 然后将其 Caption 属性值设置为“确认(&C)”。

(18) 在窗体 frmResults 的代码窗口中编写命令按钮控件 cmdConfirm 的 Click 事件过程, 程序代码如下:

'当单击“确认”按钮时执行以下事件过程

```
Private Sub cmdConfirm_Click()
```

```
Unload frmSelect      '卸载选课窗体
```

```
Unload Me             '卸载当前窗体
```

```
End Sub
```

(19) 在窗体 frmResults 的代码窗口中编写命令按钮控件 cmdPrevious 的 Click 事件过程, 程序代码如下:

'当单击“上一步”按钮时执行以下事件过程

```
Private Sub cmdPrevious_Click()
```

```
Me.Hide               '隐藏当前窗体
```

```
frmSelect.Show        '显示选课窗体
```



```
End Sub
```

(20) 在窗体 `frmResults` 的代码窗口中编写该窗体的 `Activate` 事件过程，程序代码如下：

'当窗体 `frmResults` 成为活动窗体时执行以下事件过程

```
Private Sub Form_Activate()
```

```
    Dim i As Integer, sResult As String
```

```
    lblResults = "你选择了" & frmSelect.cmbMajor.Text & "专业。" & vbCrLf & _  
    "选择了以下 " & frmSelect.lstSelectedCourse.ListCount & " 门课程： "
```

```
    For i = 0 To frmSelect.lstSelectedCourse.ListCount - 1
```

```
        sResult = sResult & frmSelect.lstSelectedCourse.List(i) & vbCrLf
```

```
    Next
```

```
    txtResult.Text = sResult
```

```
End Sub
```

(21) 将窗体 `frmSelect` 文件和窗体 `frmResults` 文件分别保存为 `frmSelect.frm` 和 `frmResults.frm`，当前工程文件保存为工程 `04-03.vbp`。

程序测试

(1) 按下 `F5` 键，以运行程序。

(2) 从“专业”组合框中选择一个专业，此时“可选课程”列表框包含的项目将随之发生变化。

(3) 在“可选课程”列表框中单击一门课程并单击“>”按钮或双击想要选择的课程，使之加入“已选课程”列表框中。也可以单击“>>”按钮以选取全部课程。

(4) 若想要从“已选课程”列表框中移除某门课程，可以在该列表框中单击该课程并单击“<”按钮或双击该课程。也可以单击“<<”按钮以移除已经选择的全部课程。

(5) 单击“确定”按钮，此时将显示另一个窗体并列出示选课结果。

相关知识

1. 列表框控件

列表框 (`ListBox`) 控件用于显示项目列表，从中可以选择一项或多项。如果项目总数超过了可以显示的项目数，就自动在列表框控件上添加滚动条。

1) 列表框控件的常用属性

列表框控件的常用属性如下所述。

- **List**: 返回或设置列表框控件的列表部分的项目。该属性的值是一个字符串数组，数组的每个元素都是一个列表项目，列表框控件在程序设计时可以通过属性窗口来设置。
- **ListCount**: 返回列表框控件的列表部分项目的个数。

- **ListIndex**: 返回或设置列表框控件中当前选择项目的索引值, 该属性在程序设计时不可用。当选定列表的第一项时, ListIndex 属性值为 0; 如果未选定项目, 则 ListIndex 属性值是 -1。ListCount 属性包含项目数, 其值总是比最大的 ListIndex 属性值大 1。
- **SelCount**: 返回在列表框控件中被选中项目的数量。
- **Selected**: 返回或设置在列表框控件中的一个项目的选择状态。该属性的值是一个布尔值数组, 其项数与 List 属性值相同。在程序设计时是不可用的。
- **Sorted**: 用于指定列表框控件中的元素是否自动按照字母表顺序进行排序。
- **Style**: 用于指定列表框的样式。如果该属性的值为 0, 则呈现为标准列表框; 如果该属性的值为 1, 则呈现为复选框式列表框, 每一个文本项的边上都有一个复选框, 可以选择多项。

2) 列表框控件的常用方法

在程序设计时可以利用 List 属性将项目添加到列表框控件的列表中, 按下 Ctrl+Enter 组合键换行添加一个新项目, 按下 Enter 键完成项目添加。在程序运行时可以通过调用以下方法来添加或删除项目。

① **AddItem**: 将项目添加到列表框控件中, 语法格式如下:

```
oList.AddItem item, index
```

其中, oList 表示列表框控件; 参数 item 为字符串表达式, 用来指定添加到该列表框控件中的项目; 参数 index 为整数, 用来指定新项目在该列表框控件中的位置。对于列表框控件中的首项, 参数 index 的值为 0。

如果所给出的参数 index 的值有效, 则 item 将被放置在列表框控件中对应的位置。如果省略参数 index, 则当将 Sorted 属性值设置为 True 时, item 将被添加到恰当的排序位置; 当将 Sorted 属性值设置为 False 时, item 将被添加到列表的末尾。

② **RemoveItem**: 从列表框控件中删除一个项目, 语法格式如下:

```
oList.RemoveItem index
```

其中, oList 表示列表框控件; 参数 index 是一个整数, 表示想要删除的项目在列表框控件中的位置。对于列表框控件中的首项, 参数 index 的值为 0。

2. 组合框控件

组合框 (ComboBox) 控件将文本框控件和列表框控件的特性结合在一起, 既可以在控件的文本框部分输入信息, 也可以在控件的列表框部分选择一项。

组合框控件的常用属性如下所述。

- **Style**: 用于设置组合框控件的样式。如果该属性的值为 0 (默认值), 则呈现为下拉式组合框, 包括一个下拉式列表和一个文本框, 可以从列表中选择项目或在文本框中输入信息。如果该属性的值为 1, 则得到一个简单组合框, 包括一个文本框和一个不能下拉的

列表，可以从列表中选择项目或在文本框中输入信息。如果该属性的值为 2，则呈现为下拉式列表，这种样式仅允许从下拉式列表中选择项目。

- **Text:** 对于将 Style 属性值设置为 0 的下拉式组合框或将 Style 属性值设置为 1 的简单组合框，返回或设置编辑域中的文本；对于将 Style 属性值设置为 2 的下拉式列表框，返回在列表框中选择的项目，返回值总与表达式 List(ListIndex)的返回值相同。

与列表框控件相同，在程序设计时可以使用 List 属性将项目添加到组合框控件的列表中。

若想要在程序运行时添加或删除组合框控件中的项目，则需要使用 AddItem 或 RemoveItem 方法。利用 List、ListCount 和 ListIndex 属性可以实现对组合框控件中项目的访问。

3. Split 函数

Split 函数返回一个下标从零开始的一维数组，它包含指定数目的子字符串，语法格式如下：

```
Split(expression[, delimiter[, count[, compare]])
```

其中，参数 expression 是必选参数，是包含子字符串和分隔符的字符串表达式。如果参数 expression 是一个长度为零的字符串 ("")，则 Split 函数返回一个空数组，即没有元素和数据的数组。

参数 delimiter 是可选参数，用于标识子字符串边界的字符串字符。如果忽略该参数，则使用空格字符 (" ") 作为分隔符。如果参数 delimiter 是一个长度为零的字符串 ("")，则返回的数组仅包含一个元素，即完整的 expression 字符串。

参数 count 是可选参数，用于指定要返回的子字符串数，-1 表示返回所有的子字符串。

参数 compare 是可选参数，其值是一个数字，表示在判别子字符串时使用的比较方式，-1 表示使用 Option Compare 语句中的设置值执行比较；0 表示执行二进制比较；1 表示执行文字比较；2 仅用于 Microsoft Access，表示基于数据库中的信息执行比较。